11g. Multithreading Packages

Martin Alfaro
PhD in Economics

INTRODUCTION

Parallelizing code may seem straightforward at first glance. However, once we start delving into its subtleties, it's rapidly revealed that an effective implementation can be a daunting task. Naive implementations can lead to various issues, including performance problems like suboptimal load balancing or false sharing, and more severe concerns such as data races. Furthermore, even if we master the necessary skills for a correct implementation, manual parallelization often impairs the readability and maintainability of code.

To assist users in overcoming these obstacles, several packages for parallelization have emerged. These tools aim to simplify the implementation of multithreading, allowing users to leverage its benefits without grappling with low-level intricacies. In this section, we'll present a few of these packages. In particular, the focus will be on those that facilitate the application of multithreading to embarrassingly parallel problems and reductions.

The first package we explore is `OhMyThreads`. This offers a collection of high-level functions and macros that help developers parallelize operations with minimal effort. For instance, it eliminates the need to manually partition tasks and tackles performance issues like false sharing. We'll then examine the `Polyester` package. Thanks to its reduced overhead, this package excels at parallelizing workloads involving small objects. Finally, we revisit the `LoopVectorization` package to introduce the `@tturbo` macro, which combines the benefits of SIMD instructions with multithreading.

> ⚠️ **Warning!** - Loaded Package
>
> All the scripts below assume you've executed `using Base.Threads`. This allows direct access to macros like `@threads` and `@spawn`.
>
> Furthermore, we assume your Julia session is running with more than one thread available. Refer to these instructions to enable multithreading.

PACKAGE "OHMYTHREADS"

`OhMyThreads` is designed as a user-friendly package for seamlessly applying multithreading. Consistent with its minimalist approach, it introduces only a handful of essential functionalities that could easily belong to Julia's `Base`. Despite its simplicity, the package covers a wide range of

operations, including reductions.

The package's primary goal is to make parallelization widely accessible, including users without deep expertise in multithreading. Indeed, its functions and macros automatically address common pitfalls, such as data races and performance issues like false sharing.

To achieve broad applicability, `OhMyThreads` extends familiar `Base` operations into the multithreaded domain through a set of higher-order functions. Each parallelized variant follows a simple naming convention: the original function name prefixed with `t`. For example, the parallel counterpart to `map` becomes `tmap` in the package. At the same time, the package remains flexible by integrating with `ChunkSplitters`, allowing users to fine-tune how work is distributed across tasks. This customization is controlled through two keyword arguments: `nchunks` (or equivalently `ntasks`) to specify the number of subsets in the partition, and `chunksize` to define the number of elements in each partition.

> ⚠ **Warning!**
> All the scripts below assume you've executed `using OhMyThreads` to load the package.

PARALLEL MAPPING

`OhMyThreads` provides `tmap` as a multithreaded analogue of Julia's `map` function. The typical and most efficient way to call it is `tmap(foo, T, x)`, where `foo` is the function applied to each element of the collection `x`, and `T` denotes the element type of the resulting array. For instance, if the computation produces a `Vector{Float64}`, then `T` should be `Float64`.

Although you can omit the type parameter and write `tmap(foo, x)`, doing so introduces a performance penalty. This slowdown arises from a subtle source: the type instability of the underlying `Task` objects used to schedule work across threads. Because the compiler can't reliably infer the output type in this case, it's forced into less efficient code paths. To avoid this and recover full performance, you should explicitly specify the output element type. In practice, rather than hard-coding `T`, it's often clearer and safer to use `eltype(x)`, which ensures that the output array mirrors the element type of the input collection.

The package also offers an in-place variant `tmap!`, whose syntax is `tmap!(foo, output, x)`. Since the destination array is provided explicitly, the function already knows the output type, so there's no need to supply `T` to prevent performance issues.

The examples that follow illustrate how to use both `tmap` and `tmap!`. For context, we also include the corresponding results from `map` and `map!`, which serve as single-threaded baselines.

MULTITHREADED MAP

```
x                       = rand(1_000_000)


foo(x)                  =  map(log, x)
foo_parallel1(x)        = tmap(log, x)
foo_parallel2(x)        = tmap(log, eltype(x), x)
```

```
julia> foo($x)
  3.407 ms (3 allocations: 7.629 MiB)
julia> foo_parallel1($x)
  2.548 ms (251 allocations: 27.197 MiB)
julia> foo_parallel2($x)
  336.252 μs (169 allocations: 7.643 MiB)
```

MULTITHREADED MAP!

```
x                       = rand(1_000_000)
output                  = similar(x)

foo!(output,x)          =  map!(log, output, x)
foo_parallel!(output,x) = tmap!(log, output, x)
```

```
julia> foo($output, $x)
  3.394 ms (0 allocations: 0 bytes)
julia> foo_parallel!($output, $x)
  333.281 μs (163 allocations: 13.516 KiB)
```

`tmap` also allows us to control the work distribution among tasks through the keyword arguments `nchunks` and `chunksize`. These options are internally implemented via the package `ChunkSplitters`.

`tmap` also gives you fine-grained control over how work is divided among tasks through the keyword arguments `nchunks` and `chunksize`. These options rely internally on the `ChunkSplitters` package. Specifically, `nchunks` controls the number of subsets in the partition, while `chunksize` sets the number of elements per task. Note that `nchunks` and `chunksize` are mutually exclusive options, so that only one of them can be used at a time.

To illustrate the use `nchunks`, we'll set its value equal to `nthreads()`. By setting a number of chunks equal to the number of worker threads, we're adopting an even distribution among tasks, similar to how `@threads` operates. To replicate the same behavior with `chunksize`, we'll make use of the floor division operator `÷`. This is a binary operator that rounds a division down to the nearest integer towards zero. [1]

NUMBER OF CHUNKS

```
x      = rand(1_000_000)

foo(x) = tmap(log, eltype(x), x; nchunks = nthreads())
```

```
julia> @btime foo($x)
  341.808 μs (169 allocations: 7.643 MiB)
```

SIZE OF CHUNKS

```
x      = rand(1_000_000)

foo(x) = tmap(log, eltype(x), x; chunksize = length(x) ÷ nthreads())
```

```
julia> @btime foo($x)
  336.906 μs (177 allocations: 7.643 MiB)
```

Do-Block Syntax

When passing anonymous functions into `tmap`, the do-block syntax comes in handy for keeping code readable. It enables the creation of multi-line functions, without the need to introduce a new function before applying `tmap`.

ANONYMOUS FUNCTION

```
x = rand(1_000_000)

function foo(x)

    output = tmap(a -> 2 * log(a), x)



    return output
end
```

DO-BLOCK SYNTAX

```
x = rand(1_000_000)

function foo(x)

    output = tmap(x) do a
                 2 * log(a)
             end

    return output
end
```

ARRAY COMPREHENSIONS

`OhMyThreads` also provides a parallel implementation of **array comprehensions**. Unlike the standard syntax of array comprehensions, the version from `OhMyThreads` combines a **generator** with a multithreaded variant of `collect` named `tcollect`.

As with `tmap`, specifying the output's element type is optional, abut necessary to avoid performance losses. The recommended syntax is therefore `tcollect(T, <generator>)`, where `T` denotes the element type of the output. A common practice is to use `eltype(x)` as `T`, with `x` being the variable iterated over in the generator. This ensures that the output type matches the input collection.

MULTITHREADED ARRAY COMPREHENSION

```
x                = rand(1_000_000)
output           = similar(x)

foo(x)           = [log(a) for a in x]
foo_parallel1(x) = tcollect(log(a) for a in x)
foo_parallel2(x) = tcollect(eltype(x), log(a) for a in x)
```

```
julia> foo($x)
  3.552 ms (3 allocations: 7.629 MiB)
julia> foo_parallel1($x)
  1.549 ms (251 allocations: 27.197 MiB)
julia> foo_parallel2($x)
  335.952 µs (169 allocations: 7.643 MiB)
```

REDUCTIONS AND MAP-REDUCTIONS

`OhMyThreads` additionally provides multithreaded counterparts to `reduce` and `mapreduce`, respectively referred to as `treduce` and `tmapreduce`. These functions automatically handle the inherent race conditions arising in reductions. They also address performance issues such as **false sharing**. Unlike `tmap`, these functions are capable of achieving optimal performance without the need to explicitly specify an output type.

MULTITHREADED MAP

```
x              = rand(1_000_000)

foo(x)         =  reduce(+, x)
foo_parallel(x) = treduce(+, x)
```

```
julia> foo($x)
  97.204 µs (0 allocations: 0 bytes)
julia> foo_parallel($x)
  22.221 µs (158 allocations: 13.891 KiB)
```

MULTITHREADED ARRAY COMPREHENSION

```
x               = rand(1_000_000)

foo(x)          =  mapreduce(log, +, x)
foo_parallel(x) = tmapreduce(log, +, x)
```

```
julia> foo($x)
  3.464 ms (0 allocations: 0 bytes)
julia> foo_parallel($x)
  361.760 µs (158 allocations: 13.891 KiB)
```

FOREACH AS A FASTER OPTION FOR MAPPINGS

`OhMyThreads` also offers a multithreaded version of `foreach` called `tforeach`. Since we haven't covered the single-threaded version `foreach`, we begin by presenting it. The function follows a syntax identical to `map`, and is usually implemented using a do-block syntax.

FOR-LOOP

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  3.521 ms (3 allocations: 7.629 MiB)
```

FOREACH (ANONYMOUS FUNCTION)

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    foreach(i -> output[i] = log(x[i]), eachindex(x))



    return output
end
```

```
julia> foo($x)
  3.374 ms (3 allocations: 7.629 MiB)
```

FOREACH (DO-BLOCK SYNTAX)

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    foreach(eachindex(x)) do i
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  3.374 ms (3 allocations: 7.629 MiB)
```

Despite the similarities between `tforeach` and `tmap`, the former is more performant. Furthermore, it doesn't incur a performance penalty when the output type isn't specified.

FOR-LOOP (SINGLE-THREADED)

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  3.442 ms (3 allocations: 7.629 MiB)
```

TMAP

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    tmap(eachindex(x)) do i
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  1.955 ms (260 allocations: 34.827 MiB)
```

TMAP (WITH OUTPUT'S TYPE)

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    tmap(eltype(x), eachindex(x)) do i
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  575.555 μs (179 allocations: 15.273 MiB)
```

TFOREACH

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    tforeach(eachindex(x)) do i
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  339.212 μs (164 allocations: 7.643 MiB)
```

Similar to `tmap`, the `tforeach` function offers the keyword arguments `nchunks` and `chunksize`, which control the workload distribution among worker threads. To illustrate, we use a work distribution analogous to the one used above for `tmap`.

NUMBER OF CHUNKS

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    tforeach(eachindex(x); nchunks = nthreads()) do i
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  339.700 μs (164 allocations: 7.643 MiB)
```

SIZE OF CHUNKS

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    tforeach(eachindex(x); chunksize = length(x) ÷ nthreads()) do i
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  341.067 μs (172 allocations: 7.643 MiB)
```

PARALLEL FOR-LOOPS

`OhMyThreads` also offers the possibility of directly parallelizing for-loops. This is done by prefixing for-loops with the `@tasks` keyword. The parallelization process can be refined by combining `@tasks` with the `@set` macro. This macro must appear as the first statement inside the for-loop body, giving you control over how work is partitioned. In particular, we can choose how many tasks to spawn using the `nchunks` option (or equivalently `ntasks`). Alternatively, we can specify the size of each chunk through the option `chunksize`, letting the number of tasks to be spawned be implicitly determined.

@TASKS

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    @tasks for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  339.714 μs (164 allocations: 7.643 MiB)
```

WITH NUMBER OF CHUNKS

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    @tasks for i in eachindex(x)
        @set nchunks = nthreads()                    # number of tasks spawned equal to the
number of threads
            output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  339.809 μs (164 allocations: 7.643 MiB)
```

WITH SIZE OF CHUNKS

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    @tasks for i in eachindex(x)
        @set chunksize = length(x) ÷ nthreads()     # size that evenly distributes work
            output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  339.564 μs (172 allocations: 7.643 MiB)
```

The `@set` macro also enables parallel reductions. This is achieved by the `reducer` option, which requires specifying the binary operator used for the reduction (e.g., `reducer = +` or `reducer = max`). Importantly, reductions in this package don't follow the usual pattern of updating an accumulator with operators like `+=`. Instead, `@tasks` assumes that the last expression of the for-loop is the value to be reduced.

SINGLE-THREAD REDUCTION

```
x = rand(1_000_000)

function foo(x)
    output = 0.0

    for i in eachindex(x)
        output += log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  3.417 ms (0 allocations: 0 bytes)
```

PARALLEL REDUCTION

```
x = rand(1_000_000)

function foo(x)
    output = @tasks for i in eachindex(x)
                @set reducer = +
                    log(x[i])
            end


    return output
end
```

```
julia> foo($x)
  350.873 μs (162 allocations: 13.609 KiB)
```

POLYESTER: PARALLELIZATION FOR SMALL PROGRAMS

> ⚠ **Warning!**
>
> All the scripts below assume you've executed `using Polyester` to load the package

One key limitation of multithreading is the overhead introduced by the creation and scheduling of tasks. For smaller computational workloads, this overhead can outweigh any potential performance gain, rendering parallelization detrimental. As a result, multithreading is typically reserved for objects large enough to justify the cost.

The `Polyester` package addresses this limitation by providing a low-overhead implementation of for-loops. Its approach makes it possible to parallelize operations on objects that, otherwise, would be deemed too small to benefit from multithreading. To use `Polyester`, we simply prefix the for-loop with the `@batch` macro.

To illustrate its benefits, let's consider a for-loop with 500 iterations, a relatively low number for applying multithreading. The first tab below shows that an approach based on `@threads` is slower than its single-threaded variant. In contrast, `Polyester` achieves a comparable performance to the single-threaded variant, even with such a small workload.

FOR-LOOP (SINGLE-THREADED)

```
x = rand(500)

function foo(x)
    output = similar(x)

    for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  1.658 μs (3 allocations: 4.008 KiB)
```

@THREADS

```
x = rand(500)

function foo(x)
    output = similar(x)

    @threads for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  9.819 μs (125 allocations: 16.633 KiB)
```

@BATCH

```
x = rand(500)

function foo(x)
    output = similar(x)

    @batch for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  1.134 μs (3 allocations: 4.008 KiB)
```

For larger workloads, it's worth noting that `Polyester` may not consistently outperform or underperform other multithreading approaches. Ultimately, performance will depend on the specifics of the computation and data involved. In such cases, you must benchmark the different implementations.

REDUCTIONS

`Polyester` also supports reduction operations. These can be implemented by prepending the for-loop with the expression `@batch reduce=(<tuple containing operation and variable reduced>)`. The implementation has been designed to avoid common pitfalls of reductions, such as data races and false sharing. This ensures both correctness and performance.

SINGLE THREADED

```
x = rand(250)

function foo(x)
    output = 0.0

    for i in eachindex(x)
        output += log(x[i])
    end

    return output
end
```

```
julia> @btime foo($x)
  766.042 ns (0 allocations: 0 bytes)
```

@BATCH

```
x = rand(250)

function foo(x)
    output = 0.0

    @batch reduction=( (+, output) ) for i in eachindex(x)
        output += log(x[i])
    end

    return output
end
```

```
julia> @btime foo($x)
  582.652 ns (0 allocations: 0 bytes)
```

It's possible to incorporate more than one reduction operation per iteration, as demonstrated below.

SINGLE THREADED

```
x = rand(250)

function foo(x)
    output1 = 1.0
    output2 = 0.0

    for i in eachindex(x)
        output1 *= log(x[i])
        output2 += exp(x[i])
    end

    return output1, output2
end
```

```
julia> @btime foo($x)
  1.374 μs (0 allocations: 0 bytes)
```

@BATCH

```
x = rand(250)

function foo(x)
    output1 = 1.0
    output2 = 0.0

    @batch reduction=( (*, output1), (+, output2) ) for i in eachindex(x)
        output1 *= log(x[i])
        output2 += exp(x[i])
    end

    return output1, output2
end
```

```
julia> @btime foo($x)
  697.073 ns (0 allocations: 0 bytes)
```

@BATCH (EQUIVALENT)

```
x = rand(250)

function foo(x)
    output1 = 1.0
    output2 = 0.0

    @batch reduction=( (*, output1), (+, output2) ) for i in eachindex(x)
        output1  = output1 * log(x[i])
        output2  = output2 + exp(x[i])
    end

    return output1, output2
end
```

```
julia> @btime foo($x)
  700.439 ns (0 allocations: 0 bytes)
```

LOCAL VARIABLES

Unlike `@threads`, `Polyester` treats variables as local per iteration.

SINGLE THREADED

```
function foo()
    out  = zeros(Int, 2)
    temp = 0

    for i in 1:2
        temp   = i; sleep(i)
        out[i] = temp
    end

    return out
end
```

```
julia> foo()
2-element Vector{Int64}:
 1
 2
```

@THREADS (OK)

```
function foo()
    out  = zeros(Int, 2)


    @threads for i in 1:2
        temp   = i; sleep(i)
        out[i] = temp
    end

    return out
end
```

```
julia> foo()
2-element Vector{Int64}:
 1
 2
```

@THREADS (DATA RACE)

```
function foo()
    out  = zeros(Int, 2)
    temp = 0

    @threads for i in 1:2
        temp   = i; sleep(i)
        out[i] = temp
    end

    return out
end
```

```
julia> foo()
2-element Vector{Int64}:
 1
 1
```

@BATCH (OK)

```
function foo()
    out  = zeros(Int, 2)
    temp = 0

    @batch for i in 1:2
        temp   = i; sleep(i)
        out[i] = temp
    end

    return out
end
```

```
julia> foo()
2-element Vector{Int64}:
 1
 2
```

SIMD + MULTITHREADING

> ⚠ **Warning!**
>
> All the scripts below assume you've executed `using LoopVectorization` to load the package

We've already covered the package `LoopVectorization` in the context of SIMD instructions. We now revisit this package to demonstrate its ability to combine SIMD with multithreading. The parallelization is achieved through its integration with `Polyester`.

The primary way to simultaneously implement SIMD and multithreading is via the `@tturbo` macro. This provides a parallelized version of `@turbo` in for-loops. Unlike the `@threads` macro, where the application of SIMD optimizations is left to the compiler's discretion, `@tturbo` enforces its application.

To illustrate the benefits of `@tturbo`, let's consider an example where SIMD isn't applied automatically by `@threads`, even when the operation is well-suited for this purpose.

SINGLE THREADED

```
x = BitVector(rand(Bool, 100_000))
y = rand(100_000)

function foo(x,y)
    output = similar(y)

    for i in eachindex(x)
        output[i] = ifelse(x[i], log(y[i]), y[i] * 2)
    end

    output
end
```

```
julia> foo($x,$y)
  608.400 μs (3 allocations: 781.320 KiB)
```

@THREADS

```
x = BitVector(rand(Bool, 100_000))
y = rand(100_000)

function foo(x,y)
    output = similar(y)

    @threads for i in eachindex(x)
        output[i] = ifelse(x[i], log(y[i]), y[i] * 2)
    end

    output
end
```

```
julia> foo($x,$y)
  82.093 μs (125 allocations: 793.945 KiB)
```

@TTURBO

```
x = BitVector(rand(Bool, 100_000))
y = rand(100_000)

function foo(x,y)
    output = similar(y)

    @tturbo for i in eachindex(x)
        output[i] = ifelse(x[i], log(y[i]), y[i] * 2)
    end

    output
end
```

```
julia> foo($x,$y)
  64.067 µs (3 allocations: 781.320 KiB)
```

The `@tturbo` macro also applies to broadcast operations. Offering this functionality is particularly valuable, as no built-in macro currently exists to parallelize broadcast expressions.

While applying `@tturbo` in a for-loop form often yields higher performance, the broadcast variant offers a much simpler and more concise syntax. The example below illustrates the improvement in readability stemming from this approach.

@TTURBO (FOR-LOOP)

```
x      = rand(1_000_000)

function foo(x)
    output = similar(x)

    @tturbo for i in eachindex(x)
        output[i] = log(x[i]) / x[i]
    end

    return output
end
```

```
julia> foo($x)
  500.463 µs (3 allocations: 7.629 MiB)
```

@TTURBO (BROADCASTING)

```
x      = rand(1_000_000)

foo(x) = @tturbo log.(x) ./ x
```

```
julia> foo($x)
  499.715 µs (3 allocations: 7.629 MiB)
```

FLOOPS: PARALLEL FOR-LOOPS (*OPTIONAL*)

> ⚠ **Warning!**
> All the scripts below assume you've executed `using FLoops` to load the package

We conclude this section with a brief overview of the package `FLoops`. The presentation is labeled as optional since usage beyond simple applications may require some workarounds. In addition, the package doesn't appear to be actively maintained.

The primary macro offered by the package is `@floop`, exclusively designed for use with for-loops. An example of its application is provided below.

FOR-LOOP (SINGLE-THREADED)

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  3.450 ms (3 allocations: 7.629 MiB)
```

@FLOOP

```
x = rand(1_000_000)

function foo(x)
    output = similar(x)

    @floop for i in eachindex(x)
        output[i] = log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  439.872 µs (158 allocations: 7.645 MiB)
```

`@floop` can also be used for reductions. This requires placing `@reduce` at the beginning of the line containing the reduction operation. The macro addresses the inherent **data race** associated with parallelization and additionally prevents false sharing.

FOR-LOOP (SINGLE-THREADED)

```
x = rand(1_000_000)

function foo(x)
    output = 0.0

    for i in eachindex(x)
        output += log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  3.414 ms (0 allocations: 0 bytes)
```

FALSE SHARING

```
x = rand(1_000_000)

function foo(x)
    chunk_ranges    = index_chunks(x, n=nthreads())
    partial_outputs = zeros(length(chunk_ranges))

    @threads for (i,chunk) in enumerate(chunk_ranges)
        for j in chunk
            partial_outputs[i] += log(x[j])
        end
    end

    return sum(partial_outputs)
end
```

```
julia> foo($x)
  1.468 ms (124 allocations: 13.250 KiB)
```

@FLOOP

```
x = rand(1_000_000)

function foo(x)
    output = 0.0

    @floop for i in eachindex(x)
        @reduce output += log(x[i])
    end

    return output
end
```

```
julia> foo($x)
  416.137 μs (252 allocations: 17.516 KiB)
```

FOOTNOTES

1. For example, `5 ÷ 3` would return `1`.